

Programming Assignment 4

Semantic Analysis using Reference Attribute Grammars

The goal of this assignment is to understand how semantic analysis can be implemented using *reference attribute grammars* (RAGs). RAGs add *computed* properties called *attributes* to abstract syntax trees (ASTs). For the language SimpliC, you will use RAGs and:

- Implement name analysis (using *synthesized and inherited attributes*)
 - Implement type checking (using *non-terminal attributes* (NTAs) for primitive types)
 - Implement error checking (using a *collection attribute*)
- Tasks that you need to do in advance are marked with a black triangle, like this. **Bring your results, e.g., AST drawings, to the lab.**
- ▷ Additional tasks that need to be completed are marked with a hollow triangle, like this.

1 The CalcRAG demo

The CalcRAG demo project illustrates how to implement semantic analysis using reference attribute grammars (RAGs). The previous assignment implemented name analysis using imperative methods and an explicit symbol table. No symbol table is needed when using RAGs, instead the information is stored in the AST.

- Download the CalcRAG demo. Run the tests and see that they pass.

1.1 Abstract grammar

The abstract grammar makes a difference between declarations of names (`IdDecl`) and uses of names (`IdUse`).

- Study the abstract grammar for CalcRAG (`src/jastadd/calc.ast`). Draw an AST for the following example program. Draw the tokens as boxes attached to the nodes, and fill in their names and values, e.g., `ID=a`.

```
let a = 1.5 in
  a * 2.0
end
```

1.2 Name Analysis

Name bindings are represented by an attribute `IdUse.decl` which refers to the appropriate `IdDecl` node. The name analysis module (`src/jastadd/NameAnalysis.jrag`) implements `IdUse.decl`, using the Lookup pattern, with the helper attributes `lookup` and `localLookup` that search for declarations. To implement declaration-before-use, the `localLookup` attribute has an integer argument `until` which is used to restrict the declaration search to an initial part of the declaration list of the `Let` construct.¹

Undeclared names are handled using the Null object pattern, and are bound to the object `unknownDecl()` which is a specific `IdDecl` object defined in another aspect. The attribute `isUnknown()` can be used on `IdDecl` objects to find out if it is the `unknownDecl()` object or not.

¹Note that the declarations in a `Let` are called **Bindings** in the abstract grammar. This is because in the Calc language, these declarations bind a value to a variable, like in functional programming.

To identify multiple declarations of the same name, an attribute `IdDecl.isMultiplyDeclared` is used. It checks if a lookup of the name results in another `IdDecl` node than the `IdDecl` itself.

Another error that can occur is that a variable is circularly defined, i.e., it is used to define its own value. To identify this situation, an attribute `IdUse.isCircular` is used. It checks if the `IdUse` occurs in the defining expression of its declaration.

1.2.1 A simple name binding

- Study the name analysis module. Extend your AST drawing by adding a box for the `decl` attribute of the `IdUse` node. In order to evaluate `decl`, a couple of additional attributes will be evaluated as well: `lookup("a")` for the `IdUse` node, and `localLookup("a")` for the `Let` node. Draw these attributes as boxes too. Then think about what the values of the attributes will be, and add them to your drawing (draw reference values as an arrow from the attribute box to the appropriate node).

1.2.2 Nesting and undeclared uses

Note how nested scopes are implemented: the equation for `lookup` in class `Let` delegates to its own `lookup` attribute in case the declaration is not found locally.

- Consider the program below that contains two undeclared uses: the first use of `b` (in `a = b`) is not within the scope of the subsequent declaration of `b`, and there is no declaration of `c`. Draw the AST and the attributes needed to evaluate the `decl` attributes of these two uses. The `unknownDecl()` object can be drawn as an object separate from the AST. We will see in the next section how it is implemented. You will find that by following the equations, you will need to evaluate the `lookup` attribute for the `Let` node, and the `decl` attributes will refer to the `unknownDecl()` object.

```
let
  a = b
  b = 3.0
in
  a * c
end
```

1.2.3 Multiply declared variables

- Consider the program below that contains two declarations of the same name, within the same `Let` construct. Draw the AST, and the `isMultiplyDefined` attributes of the two declarations. Extend the drawing with the attributes needed to evaluate the `isMultiplyDefined` attributes. Note that only the second declaration is considered multiply defined. Consider the use of the variable `a` inside the body. Which `IdDecl` will it be bound to?

```
let
  a = 1.0
  a = 2.0
in
  a
end
```

1.2.4 Circularly defined variables

- Consider the program below with a circular definition: Inside the declaration of **a**, there is a use of **a** (in the expression **a * b**). Draw the AST for this program. Draw the **decl** attributes of the three **IdUse** nodes. Each **IdUse** node has an attribute **isCircular**. Draw these attributes and compute their values.

```
let a =  
    let b = 2.0 in a * b end  
in  
    a * 3.0  
end
```

1.3 Unknown declarations

Support for unknown (missing) declarations, is implemented in the module `src/jastadd/UnknownDecl.jrag`. Unknown declarations are implemented using the *Null object pattern*²: an object is used instead of the null value. The abstract grammar contains a subclass to **IdDecl** called **UnknownDecl**:

```
UnknownDecl: IdDecl;
```

All unknown declarations are represented by the same object. To create such an object, we use a *non-terminal attribute* (NTA), that is, an attribute whose value is a *new subtree*. An ordinary reference attribute refers to an existing tree node, but an NTA, in contrast, refers to a new subtree created by its defining equation. NTAs are used for creating AST nodes that were not created by the parser. The following code adds an **UnknownDecl** as an NTA to the **Program** node.

```
syn nta UnknownDecl Program.unknownDecl() = new UnknownDecl("<unknown>");
```

Later in this assignment, we will use NTAs for representing predefined functions and primitive types in the SimpliC language.

To give easy access to the **UnknownDecl** object from other places in the AST, we will use a common pattern called *Root Attribute*, which gives access to a particular attribute from anywhere in the AST. This is accomplished by declaring the attribute as an inherited attribute for **ASTNode** (thereby giving all nodes access to it), and by an equation in the root that propagates the value to all its children. Through the JastAdd broadcast mechanism, the value is propagated throughout the complete subtree:

```
inh UnknownDecl ASTNode.unknownDecl();  
eq Program.getChild().unknownDecl() = unknownDecl();
```

This attribute is used in the name analysis. We can also see another attribute **isUnknown**. This attribute is simply defined as:

```
syn boolean IdDecl.isUnknown() = false;  
eq UnknownDecl.isUnknown() = true;
```

The attribute **isUnknown** illustrates the benefits of using the null object pattern: we can define attributes that are implemented differently for normal objects (**IdDecls**) and the null object (**UnknownDecl**).

- Add the **Program.unknownDecl** NTA attribute to one of your previous AST drawings.

²See http://en.wikipedia.org/wiki/Null_Object_pattern

1.4 Collecting errors

Compile-time errors are implemented in the module `src/jastadd/Errors.jrag`. The errors are represented by an attribute `Program.errors` that is a set of error messages. The `errors` attribute is declared as a *collection attribute*:

```
coll Set<ErrorMessage> Program.errors() [new TreeSet<ErrorMessage>()] with add root Program;
```

The value of a collection attribute is defined by *contribution* rules. The contributions are automatically collected by the attribute evaluator, traversing the AST and adding each contribution to the collection. In this case, the initial value of the collection is a `new TreeSet<ErrorMessage>()`, and the contributions are added with the method `add` on `TreeSet`. The type `ErrorMessage` contains an error message and a line number. The set is sorted by the line number.

The final part of the collection declaration, `root Program`, restricts the traversal to the subtree rooted by the `Program` node containing the `errors` attribute. In this case, the whole AST will be traversed, since `Program` is also the root of the whole AST.

The compiler checks three kinds of errors, each defined using a contribution: undeclared variables, multiply declared variables, and circularly defined variables. Consider the example for undeclared variables below. The contribution specifies

- what error message to add (in case there actually is an error)
- when to add the error message (there is an undeclared variable)
- what collection attribute it should be added to (`Program.errors()` in this case)
- which node contains the collection attribute (`program` in this case)

```
IdUse contributes error("symbol '" + getID() + "' is not declared")
when decl().isUnknown()
to Program.errors() for program();
```

The `program` attribute is an inherited attribute again implemented using the Root Attribute pattern, that propagates a reference to the root down to all nodes in the AST. The `error` method creates an object of the type `ErrorMessage` and sets the line number.

A key benefit of using collection attributes is that contributions can be spread out in different aspects. This way, if the language is extended, or if new analyses like type checking are added, new kinds of compile-time errors can easily be added. JastAdd will automatically collect all contributions to a collection attribute.

- Study the module for compile-time errors, and make sure you understand how the collection of error messages works. Add a test case with a Calc program that contains a cycle in a declaration (where x is defined in terms of x), and verify that the compiler complains.

1.5 Demand evaluation and attribute caching

Attributes in JastAdd are *evaluated on demand*, that is, computed when they are accessed. When an attribute is accessed, the corresponding equation is executed. If an attribute is evaluated several times, the equation will be executed several times, which is not practical for large compilers. To speed up the execution, JastAdd can automatically cache attributes, that is, storing the value of an accessed attribute for further accesses. This can be done by adding the keyword `lazy`, as the following code illustrates.

```
syn lazy IdDecl IdUse.decl() = lookup(getID());
```

This statement will add a (Java) field of type `IdDecl` in the generated `IdUse` class, which is used to cache the value of the attribute.

Caching all attributes usually gives sufficient performance. Tuning the caching behavior gives rather modest speed and memory improvements, but can be interesting for production compilers. In the assignments, we will therefore always cache all attributes. This is achieved by giving JastAdd the option `cache="all"` (see `build.xml`), so we will never need to use the `lazy` modifier in the specifications.

1.6 Generated Java code

JastAdd takes a set of aspects and abstract grammars and generates Java classes. Each attribute is translated to a method in the class the attribute is declared on. For example, for the attribute `IdUse.decl`, a method `decl()` is generated in the class `IdUse`. JastAdd generates different code depending on the kind of attribute. The simplest attributes are synthesized attributes. These correspond to ordinary methods with extra code that adds caching of attributes and a circularity check. An attribute instance that depends on itself is invalid, unless it is explicitly declared as `circular`, and JastAdd throws an exception during execution of such an attribute. Note that an attribute can access the same attribute of another node without necessarily being circular. For example, the `lookup` attribute accesses `lookup` attributes of other nodes. Some problems are naturally circular, and in those cases, attributes on the cycle can be explicitly declared as `circular`. JastAdd will then solve the equations using iteration, running the equations over and over until they are solved. More on circular attributes in the next assignment.

- Look at the method for the `IdUse.decl` attribute in the generated code. Check that JastAdd generates code for caching this attribute.
- Write an attribute that is circular (without using the `circular` keyword). Invoke the attribute and verify that an exception is thrown.

2 SimpliC

You will now extend your compiler for SimpliC from assignment 2 or 3 by semantic analysis using RAGs. If you are extending the assignment 3 version, you can simply create a new main program that does not call the visitors or aspects from assignment 2.

- Make sure the code you start from builds and tests correctly. Make a copy of your code, so that you can return to a well-defined state, in case you run into a blind alley and want to discard your changes. (Or better, use a version control system.)

2.1 Name analysis and error checking

- Implement name analysis for SimpliC using RAGs. The semantics regarding declaration order and shadowing should be the same as in assignment 3: both functions and variables share the same name space. All names need to be declared before use. Ignore predefined functions (`read` and `print`) for the moment. Use the Lookup pattern and the Null object pattern, as for the Calc example.
- Add error checking by using a collection attribute for errors, i.e., in the same way as in the Calc example. Report undeclared names and multiply declared names. Add suitable test cases. You can copy the `CalcRAG/src/java/TestNameAnalysis.java` testing framework to easily add name analysis test cases. As you see in the testing code, a source file is parsed into an AST, then the `program.errors()` attribute is accessed and printed to a file, which then is compared with the expected output file.

2.2 Predefined functions

The SimpliC language has two predefined functions: `read` and `print`. These functions can be represented as an NTA in the AST. In this case, we want the NTA to be a `List` node that contains one subtree for each of the functions. The following code illustrates how such an NTA can be specified.

```
syn nta List<FunctionDecl> Program.predefinedFunctions() {
    List<FunctionDecl> list = new List<FunctionDecl>();
    // Create objects of type FunctionDecl and add them to the list
    return list;
}
```

The Root Attribute pattern can be used for propagating the `predefinedFunctions` attribute to the whole AST, or to the places where it is needed.

- ▷ Extend your name analysis with support for predefined functions. Define the attribute `predefinedFunctions`. Use this attribute in the equation of the `lookup` attribute. Make sure that you have test cases that cover the use of predefined functions and that predefined functions cannot be redeclared by other functions.

2.3 Type analysis

Type analysis is about assigning types to expressions. There are three types in SimpliC: integer, boolean and the unknown type. Variables, parameters and the return type of a function can only be of integer type. This limitation exists to make the assignment easier. Arithmetic expressions are of type integer and boolean expressions are of type boolean. The statements `if` and `while` require the condition to be of type boolean. The unknown type is assigned to variable uses that refer to unknown names. An unknown type makes it easier to limit the propagation of errors – we only want to report relevant errors and not errors that are caused by other errors. The types can be modelled as the following abstract grammar fragment shows:

```
abstract Type;
IntType: Type;
BoolType: Type;
UnknownType: Type;
```

- ▷ Implement the primitive types using NTAs, in the same way as you did for the unknown name. Use the Root Attribute pattern to propagate references to these objects throughout the program.
- ▷ Add the attribute `type` to both expressions and `IdDecl`, returning the type of the expression or declaration. Give appropriate equations for the attributes.

We now want to report type errors. This can be done by adding a contribution to the `errors` attribute. One approach is to add an inherited attribute `expectedType` to expressions that describes the type to be expected. The contribution can then compare the expected type with the actual type, and if they are incompatible, an error is reported. It is also useful to add an attribute `compatibleType` on `Type` that takes another type as a parameter and compares the two types and tells if they are compatible. To limit the propagation of type errors, the unknown type should be compatible with all other types.

- ▷ Add the attributes `expectedType` and `compatibleType`. Add a contribution that contributes an error message if two types are incompatible.

2.4 More errors

We will now report more errors.

- ▷ Since variables and functions share the same name space, a variable use can refer to a function and a function call can refer to a variable. Report these errors. Add two inherited attributes `isVariable` and `isFunction` to `IdDecl` and use them in the contributions.
- ▷ Report if the number of actual arguments in a function call does not match with the number of formal parameters of that function. You can add an attribute `function` on `IdDecl` to access the function declaration, and from there the number of formal parameters.

3 Optional Challenges

When you have completed the compulsory part of the assignment you may try one or more of these optional challenges:

- Add boolean literals (`true` and `false`)
- Add the type `void` that can be used as the return type of a function
- Add the type `bool` as a type that can be used for return types, variables and parameters. Implement this by defining an attribute `lookupType` that defines name rules for types (that is, types have a separate name space).